<?xml version="1.0" encoding="UTF-8"?>

<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	
	<xsl:include href="../admin.xsl" />
	<xsl:include href="../cp/path.xsl" />
	
	<!-- Запускаем рендер страницы -->
	<xsl:template match="/data">
		<html>
		
		<head>
			<title>
				<xsl:value-of select="$pageTitle" />
			</title>
			<xsl:call-template name="meta-head" />
		</head>
		<body>
			<xsl:call-template name="admin-header" />
			
			<div id="blocks">
				<div id="blocks_inner" />
			</div>
			
			<div id="content">
				<div id="content_inner">
					<xsl:apply-templates select="response" />
				</div>
			</div>
			
			<xsl:call-template name="footer" />
		</body>
		</html>
	</xsl:template>
	
	<xsl:template match="response">
		<xsl:param name="cntAdmins" select="count(object[@type='user' and @group = 'admin'])" />
		<xsl:param name="cntModerators" select="count(object[@type='user' and @group = 'moderator'])" />
		
		<div id="path">
			<xsl:apply-templates select="document('xmodule://cp/getPath/0;users;groups')/data/response" mode="cp-path" />
		</div>
		
		<h1>Группы пользователей</h1>
		
		<h2>Добавить пользователя</h2>
		
		<form action="" method="post">
			<p>
				<xsl:text>Логин: </xsl:text>
				<input type="text" name="login" value="" size="20" />
				<xsl:text> </xsl:text>
				<input type="submit" name="add_admin" value="в администраторы" />
				<xsl:text> </xsl:text>
				<input type="submit" name="add_moderator" value="в модераторы" />
			</p>
			
			<p>Повторное добавление пользователя удалит его из заданной группы.</p>
		</form>
		
		<h2>Администраторы</h2>
		
		<p>
			<xsl:if test="$cntAdmins = 0">
				<xsl:text>Фантастика, не администраторов</xsl:text>
			</xsl:if>
			
			<xsl:for-each select="object[@type='user' and @group = 'admin']">
				<a>
					<xsl:attribute name="href">
						<xsl:value-of select="concat($baseURI, '/user/', property[@name='login']/value)" />
					</xsl:attribute>
					<xsl:value-of select="property[@name='name']/value" />
				</a>
				
				<xsl:if test="position() != $cntAdmins">
					<xsl:text>, </xsl:text>
				</xsl:if>
			</xsl:for-each>
		</p>
		
		<h2>Модераторы</h2>
		
		<p>
			<xsl:if test="$cntModerators = 0">
				<xsl:text>Модераторов нет</xsl:text>
			</xsl:if>
			
			<xsl:for-each select="object[@type='user' and @group = 'moderator']">
				<a>
					<xsl:attribute name="href">
						<xsl:value-of select="concat($baseURI, '/user/', property[@name='login']/value)" />
					</xsl:attribute>
					<xsl:value-of select="property[@name='name']/value" />
				</a>
				
				<xsl:if test="position() != $cntModerators">
					<xsl:text>, </xsl:text>
				</xsl:if>
			</xsl:for-each>
		</p>
	</xsl:template>
	
	
	
</xsl:stylesheet>